


PRÁCTICA 5: Gancho de grúa

1. ENUNCIADO

Analizar el comportamiento de un gancho de grúa diseñado para levantar una carga de $F=5000$ N. El gancho tiene un espesor uniforme de 5 mm y las dimensiones mostradas en la figura (en mm). Las propiedades del acero son: módulo de Young $E = 210$ GPa; módulo de Poisson $\nu = 0.3$. Obtener las deformaciones y las tensiones en la pieza.



$$r_1=12 \text{ mm}; r_2=24 \text{ mm}; r_3=36 \text{ mm}; h=10 \text{ mm}$$


2. DESCRIPCIÓN DEL MODELO

Se trata de un ejemplo similar al de la chapa perforada de la práctica 4. Se trata de un análisis bidimensional de tensión plana. Como simplificaciones, no se modelará el extremo del gancho ya que esta parte no trabaja y por tanto de esta forma se ahorra coste de modelado y de análisis. La unión entre el gancho y la cadena es una unión articulada, pero en su lugar se empostrará el extremo superior del gancho ya que carga se encuentra alineada con este extremo y por tanto esta modelización no afecta a la respuesta del gancho.

En cuanto a la malla, se refinará en el radio interior del gancho, donde se prevé que vaya a haber una concentración de tensiones. Se prestará atención a la existencia de elementos triángulo y con mala relación de aspecto, especialmente en esta zona crítica. Se visualizarán resultados de deformaciones, así como las tensiones de Von Mises en toda la pieza y especialmente en la sección crítica.


3. RESOLUCIÓN PASO A PASO

Preferences:




Preprocessor>


> Element type > Add/Edit/Delete > Add: añadir elemento tipo plane42




>Elementtype>Add/Edit/Delete>Options: tensión plana con espesor (espesor de la pieza distinta a 1)




>Real Constants>Add/Edit/Delete>Add: introducir las propiedades de la sección de la pieza (espesor 5 mm)




> **Material Props > Material Models > Structural > Linear > Elastic > Isotropic:** definir material ($E=210000\text{N/mm}^2$, $\nu=0.3$)




> **Modeling > Create > Keypoints > In active CS:** introducir keypoints por coordenadas ((0,0,0) y (6,0,0))




>Modeling>Create>lines>By Cent &Radius: crear círculos (de radio 12 con centro en el keypoint 1, y de radio 30 con centro en el keypoint 2)




>PlotCtrls> Numbering >Keypoint numbers >On: mostrar número de keypoint en la pantalla




Modeling > Create > Lines > Straight line: crear línea entre keypoints 5 y 9




> Modeling > Create > Keypoints > In active CS: introducir keypoints por coordenadas ((-6,0,0), (6,48,0) y (-6,48,0))




Modeling > Create > Lines > Straight line: crear línea entre keypoints 8 y 12, 12 y 13, 13 y 11




>Plot>Lines: mostrar líneas en la pantalla




>PlotCtrls>Numbering> Line numbers>On: mostrar número de línea en la pantalla




Modeling>Create>LinesFillet: crear radio de acuerdo entre las líneas 5 y 10, con radio 15.8




Modeling>Operate>Booleans>Add>Lines: juntar líneas 1 y 2 para crear una línea nueva




Modeling>Operate>Booleans>Partition>Lines: particionar línea 1 con línea 12 para crear líneas nuevas




Modeling>Create>LinesFillet: crear radio de acuerdo entre las líneas 2 y 16, con radio 48




Modeling>Delete>LinesOnly: eliminar líneas 6, 14 y 15




Modeling > Create > Areas > Arbitrary > By Lines: crear un área con todas las líneas




>Meshing>MeshAttributes>AllAreas: especificar propiedades de la malla del área (material 1, real constant 1, elementtype 1)



>Meshing>Sizecontrols>ManualSize>Areas>AllAreas: especificar tamaño de los elementos de la malla (tamaño 4)




>Meshing>Mesh>Areas>Free: mallar el área seleccionándola




>Plot>Lines: mostrar líneas en la pantalla




>PlotCtrls>Numbering> Line numbers>On: mostrar número de línea en la pantalla



>Loads> Define Loads>Apply>Structural>Displacement>OnLines: introducir condiciones de ligadura en la línea seleccionada (todos los grados de libertad restringidos a 0 en la línea superior del gancho)



>PlotCtrls>Numbering>Keypointnumbers>On: mostrar número de keypoint en la pantalla




>Loads> Define Loads>Apply>Structural>Force/Moment>OnKeypoints: introducir una fuerza o momento en el keypoint seleccionado (fuerza $F_y = -5000$ en el keypoint6)




Solution>

>Solve>Current LS: analiza el modelo generado en el preprocesador




General Postproc >

>PlotResults>DeformedShape: dibujar la deformada



>PlotResults>ContourPlot>Nodal Solution>Stress>von Mises Stress: dibujar las tensiones de von Mises




>Plot>Lines: mostrar líneas en la pantalla




Preprocessor>

>Meshing>ModifyMesh> Refine at>Keypoints: refinar la malla en el keypoint 3 con un nivel de refinado de 2




Solution>

>Solve>Current LS: analiza el modelo generado en el preprocesador




General Postproc>


>PlotResults>ContourPlot>Nodal Solution>Stress>von Mises Stress: dibujar las tensiones de von Mises




>Path Operations>Define Path > By Location: crear un path entre los puntos (12,0,0) y (36,0,0)




>Path Operations> Plot Paths: dibujar el path recién creado




>PathOperations>Mapontopath> Stress > von Mises SEQV: definir la variable cuyo valor a lo largo del path queremos obtener



>PathOperations>Mapontopath> Stress > Y-direction SY: definir la variable cuyo valor a lo largo del path queremos obtener



>Path Operations> Plot Path item > On Graph > SEQV y SY: dibujar en un gráfico las variables SEQV y SY



4. RESOLUCIÓN EN APDL

```

!*****análisis estructural*****
/COM,
/COM,Preferences for GUI filtering have been set to display:
/COM, Structural
!*****
!*****PREPROCESADOR*****
!*****
/PREP7
!*****tipo de elemento*****
ET,1,PLANE42
KEYOPT,1,1,0
KEYOPT,1,2,0
KEYOPT,1,3,3
KEYOPT,1,5,0
KEYOPT,1,6,0
!*****realconstant*****
R,1,5,
!*****material*****
MPTEMP,,,,,,,,
MPTEMP,1,0
MPDATA,EX,1,,210000
MPDATA,PRXY,1,,0.3
!*****geometria*****
K,1,0,0,0,
K,2,6,0,0,
CIRCLE,1,12
CIRCLE,2,30
LSTR, 5, 9
k,11,-6,0,0
k,12,6,48,0
k,13,-6,48,0
LSTR, 8, 12
LSTR, 12, 13
LSTR, 13, 11
LFILLT,5,10,15.8
LCOMB,1,2
LPTN,1,12
LFILLT,16,2,48, ,
LDELE,6
LDELE,14
LDELE,15
AL,ALL
!*****mallado*****
AESIZE,ALL,3,
AMESH,ALL
!*****condiciones de ligadura*****
DL,11, ,ALL,
    
```

```

!*****fuerza*****
FK,6,FY,-5000
FINISH
!*****
!*****PROCESADOR*****
!*****
/SOL
/STATUS,SOLU
SOLVE
FINISH
!*****
!*****POSPROCESADOR*****
!*****
/POST1
!*****dibujar deformada*****
PLDISP,1
!*****dibujar tensión de vonMises*****
PLNSOL, S,EQV, 0,1.0
!***refinar la malla en la zona de tensión máxima***
FINISH
/PREP7
KREFINE,3, , ,2,1,1,1
!*****
!*****PROCESADOR*****
!*****
/SOL
/STATUS,SOLU
SOLVE
FINISH
!*****
!*****POSPROCESADOR*****
!*****
/POST1
!*****dibujar tensión de vonMises*****
PLNSOL, S,EQV, 0,1.0
!*****definirpath en la sección crítica*****
PATH,seccrit,2,30,20,
PPATH,1,0,12,0,0,0,
PPATH,2,0,36,0,0,0,
!*****definir variables a plotear en el path*****
PDEF, ,S,EQV,AVG
PDEF, ,S,Y,AVG
!*****dibujar variables del path en un grafico*****
PLPATH,SEQV,SY
  
```

NOTA: Todas las imágenes de este documento son propias